
OMG, UML 2.0

• Around 1999 Clark, Evans, Kent started attending OMG.
• UML 2.0 started around this time.
• The 2U Submission: UML as family of languages:

• Templates.
• Package Extension.

• Tools and methods emerged for model-based language
engineering:

• Language Architectures.
• Denotational Semantics.
• Operational Semantics.

Modelling and Programming

• Aim: to merge modelling and programming.
• Tools: MMT, XMT, XMF.
• Programming language based on FP and OCL.
• Important features: meta-; reflection; OO.
• Tools for Language Engineering.

Xactium

• Clark, Evans set up in 2003.
• Developed XModeler (XMF-Mosaic) on top of XMF.
• 2003-2008.
• Clients: BAES; BT; Citi-Group; Artisan; BSkyB.

XMF

• Meta-Circular Language (like MOF and ECore).
• XCore Based on ObjvLisp.
• File based or world-state.
• Features for:

• Packages of models/programs.
• Higher-order operations.
• OCL.
• Meta-Object Prototcol (MOP).
• Language Engineering (grammars, syntax processing).
• Daemons (object listeners).
• Pattern matching.
• Code generation templates.
• Threads.
• XML processing (parsing).
• Java integration.

XModeler

• Eclipse RCP Tool.
• Layered on (and written in) XMF.
• MVC Architecture.
• File interface for building XMF applications.
• Functional interface e.g. deploy XML, check constraints.
• Clients are all extensible:

Client: XMF Command Listener.
Client: Browsing all data.
Client: Editing all data.
Client: Diagrams: class; snapshot.
Client: Text editing, HTML browsing.

Meta-language

Models
parserImport XOCL;

context Root
@Package BasicLibrary
@Class Library

@Attribute books : Set(Book) end
@Attribute readers : Set(Reader) end
@Attribute borrows : Set(Borrows) end

end
@Class Book

@Attribute title : String end
end
@Class Reader

@Attribute name : String end
end
@Class Borrows

@Attribute reader : Reader end
@Attribute book : Book end

end
end

Behaviour

Programs: XOCL

parserImport XOCL;
import BasicLibrary;

context Book
@Constructor(title) end

context Library
@Operation addBook(title:String)

let b = Book(title)
in self.books := books->including(b); b
end

end

context Library
@Operation findBook(title:String):Book

let B = books->select(b | b.title = title)
in if B->isEmpty

then null
else B->asSeq->head
end

end
end

Snapshots

context Root
@Operation test()
let S = Snapshot("library",Seq{BasicLibrary})
in S.add("l",Library());

let l = S::l
in S.add("b",l.addBook("War and Peace"));

S.add("r",l.addReader("Fred"));
S.add("b_x_r",l.addBorrows(S::r,S::b));
S

end
end

end

Defining Constraints (1)

context Library
@Constraint all_borrowed_books_belong_to_library
borrows.book->forAll(b | books->includes(b))
fail "can only borrow books in library"

end

context Library
@Constraint all_borrowing_readers_belong_to_library

borrows.reader->forAll(r | readers->includes(r))
fail "only registered readers can borrow books"

end

Defining Constraints (2)

context Library
@Constraint cannot_borrow_same_book_twice

borrows->forAll(b1 | borrows->forAll(b2 | b1 <> b2 implies b1.book <> b2.book))
fail "cannot borrow the same book twice"

end

context Library
@Constraint all_books_have_unique_titles

books->forAll(b1 | books->forAll(b2 | b1 <> b2 implies b1.title <> b2.title))
fail "books should have unique titles"

end

context Library
@Constraint all_readers_have_unique_names

readers->forAll(r1 | readers->forAll(r2 | r1 <> r2 implies r1.name <> r2.name))
fail "readers should have unique names"

end

Checking Constraints

context Root
@Operation test_illegal()
let S = Snapshot("library",Seq{BasicLibrary});

b = Book("War and Peace")
in S.add("l",Library());

S.add("r",(S::l).addReader("Fred"));
S.add("b_x_r",(S::l).addBorrows(S::r,b));
S

end
end

[1] XMF> test_illegal().checkConstraints().failures();
Seq{ConstraintReport(<Library d9ff5f>,

<Constraint all_borrowed_books_belong_to_library>,
false, can only borrow books in library)}

[1] XMF>

Syntax Classes

Quasi-Quotes

context Root
@Operation add1_exp(exp:Performable):Performable

[| 1 + <exp> |]
end

context Root
@Operation seq_exp(exps:Seq(Performable)):Performable

exps->iterate(e x = [| Seq{} |] |
[| <x>->including(<e>) |])

end

[ 1 ] XMF> add1_exp ( [ | x + y | ] ) ;
BinExp ( IntExp ( 1 ) ,+ ,BinExp ( Var ( x ) ,+ , Var ( y ) ) )
[ 1 ] XMF> seq_exp (Seq { [ | 1 | ] , [ | x | ] , true . l i f t ( ) } ) ;
CollExp ( CollExp ( CollExp ( SetExp (Seq , Seq { } ) , including , Seq{ IntExp ( 1 ) } ) , including , Seq{ Var

( x ) } ) , including , Seq{BoolExp ( true ) } )
[ 1 ] XMF> seq_exp (Seq { [ | 1 | ] , [ | x | ] , true . l i f t ( ) } ) . pprint ( ) ;
Seq{}−>including ( 1 )−>including ( x )−>including ( true )
[ 1 ] XMF>

Grammars

parserImport Parser::BNF;
parserImport XOCL;

Root::g :=
@Grammar

Start ::= i=Int o=Op j=Int {
@Case o of

"+" do i + j end
"*" do i * j end

end
}.
Op ::= ’+’ { "+" } | ’*’ { "*" }.

end;

[ 1 ] XMF> g . parseString ("1 + 2" ," Star t " ,Seq { } ) ;
3
[ 1 ] XMF>

Embedded Language Features (Usage)

context Library
@Subset all_borrowed_books_belong_to_library

borrows.book books
end

context Library
@Subset all_borrowing_readers_belong_to_library

borrows.reader readers
end

context Library
@Unique cannot_borrow_same_book_twice

borrows book
end

context Library
@Unique all_books_have_unique_titles

books title
end

context Library
@Unique all_readers_have_unique_names

readers name
end



BACKGROUND XMF XMODELER

Embedded Language Features (Def 1)

parserImport XOCL;
parserImport Parser::BNF;

context Root
@Class Unique
@Grammar
Unique ::= name=Name collection=Name field=Name {
[| @Constraint unique self.<collection>->forAll(x |

self.<collection>->forAll (y |
x <> y implies x.<field> <> y.<field>))

end.name := "UNIQUE:" + <name.lift()> |]
}.

end
end

Embedded Language Features (Def 2)
parserImport XOCL;
parserImport Parser::BNF;
import OCL;

context Root
@Class Subset
@Grammar

Subset ::= name=Name sub=Path super=Path {
[| @Constraint contained <sub>->forAll(x |

<super>->includes(x))
end.name := "CONTAINED:" + <name.lift()> |]

}.
Path ::= root=Name fields=(’.’ Name)* {

fields->iterate(field exp = Var(root) |
[| <exp>.<field> |])

}.
end

end

Generating Code

context Library
@Operation borrowsTable()

let sout = StringOutputChannel()
in @HTML(sout,0)

<TABLE border=1>
{ @Loop borrow in borrows do

[ <TR>
<TD> { borrow.reader.name } </TD>
<TD> { borrow.book.title } </TD>

</TR> ]
e_nd }

</TABLE>
end;
sout.getString()

end
end

Tooling Requirements

Adding Daemons

o.addDaemon(
@Operation(slot,new,old)

... something to do...
end)

Clients

HTML

Tool Requirements

• Take any snapshot.
• Interactively construct class instances.
• Interactively edit class instances.
• Check constraints after each modification.
• Generate Java code for the snapshot.

Tool Models

MVC Model

Demo

Availability, Documentation, Research
http://www.eis.mdx.ac.uk/staffpages/tonyclark/

XPL

